統計数理研究所彙報
第31巻 第1号 (1983)

多次元立方体の離散的ランダムパッキングによる充填率について

統計数理研究所 伊 藤 栄 明
〃 上 田 澄 江

(1983年5月 受付)

球状粒子をランダムに充填する物理的な実験は以前から行なわれている (Bernal and Mason [2], Higuti [6]). これについての計算機シミュレーションも最近は多く行なわれている (Tanemura [9] 等). ここでは球状粒子ではなく, m 次元立方体を向きをそろえてランダムに充塡していくという問題について議論する. 一次元の場合, 様々な確率論的研究があるが (Rényi [8], Dvoretzky and Robbins [5], Itoh [7]), 2次元以上の場合, 解析的な結果は得られていないようである. 辺の長さが十分長い m 次元立方体への, m 次元単位立方体によるランダムパッキングの充塡率の期待値は一次元の場合の充塡率の期待値の m 乗になるという予想がよく知られている. これは Palásti の予想といわれているが, $m=2,3,4$ の場合, 計算機実験により否定されたようである (Blaisdell and Solomon [3], [4], Akeda and Hori [1]). Blaisdell と Solomon [4] は4次元までの結果は Palasti の予想より大きく, その差は次元とともにひらいていくと述べている. 本稿においては高い次元の充塡率を考察するためにもっとも単純な場合に議論を限る. 先ず, 計算機実験によらずに数値計算により, 期待値を求める. この方法では計算時間がかかり3次元までの値しか得られない. しかし, これにより精度の高い値が得られるだけでなく計算機実験の結果を確認することができる. 次に, 計算機実験により11次元までの値を求め, 次元が十分に大である場合の充塡率について考える.

一辺の長さ $2n$ の m 次元の立方体があるとする. その中に一辺が長さ n の m 次元の立方体を大きい方の立方体に平行にランダムにつめていく. 一辺が長さ $2n$ の m 次元立方体の 2^m 個の頂点の位置は, 各要素が0あるいは $2n$ である m 次元のベクトルで表わすことができる. 次に, 各要素が0あるいは n であるベクトルを 2^m 個の頂点として持つ m 次元立方体を考える. この立方体の位置を m 次元ベクトル $\vec{0}=(0,0,\cdots,0)$, により表わすものとする. この立方体を各要素が0から n までの自然数の値をとる m 次元ベクトル $\vec{A}$ により平行移動した $(n+1)^m$ 個の立方体を考える. これらの各々を n^m 立方体とよび, その位置を $\vec{A}$ により表わすことにする. $(n+1)^m$ 個の n^m 立方体の中から等確率で $\vec{A_1}$ をえらび, 次に $\vec{A_1}$ は $\vec{A_1}$ とまじわらない n^m 立方体, すなわち $\vec{A_2}-\vec{A_1}$ の各要素の絶対値の最大値が n 以上になっている n^m 立方体,

表 1 数値計算によって求めた充塡率 ($n=2$)

次元 m	1	2	3
充塡率 E_m	0.83333333	0.70925926	0.61622176
充塡率の標準偏差	0.23570226	0.21986403	0.17860018
項の数	3	53	137911

```
      PROGRAM SAMPLE
      LCUBE  = 4
      LONG   = 2
      M      = 3
      CALL MDPACK (LCUBE, LONG, M)
      STOP
      END
C
      SUBROUTINE MDPACK (LCUBE, LONG, M)
C     THIS SUBROUTINE CULCULATES EXPECTATION OF PACKING DENSITY
C     PACKING M DIMENSIONAL CUBES WITH LENGTH 'LONG' INTO M DIMENSIONAL
C     CUBE WITH LENGTH 'LCUBE'.
C     SIZES OF ARRAYS IA(I,J,K), IC(I,J,K), NA(K), NC(K), P(K):
C         I IS GREATER THAN OR EQUAL TO M,
C         J IS GREATER THAN OR EQUAL TO (LCUBE-LONG+1)**M, AND
C         K IS GREATER THAN OR EQUAL TO (LCUBE/LONG)**M.
C
      INTEGER*2  IA(3,343,64), IC(3,343,64), NA(64), NC(64)
      REAL*16    P(64), SUMP, SUM
C
C     INITIALIZATION
      LONG1  = LCUBE - LONG + 1
      LONGM  = LONG1 ** M
      DO 20   I=1,M
      IJ     = 0
      ID     = LONG1 ** (I-1)
      DO 20   J=1,LONGM
      IJ     = IJ + 1
      IJD    = (IJ-1) / ID
      MIJ    = MOD(IJD,LONG1)
      IA(I,IJ,1) = MIJ
   20 IC(I,IJ,1) = MIJ
      L      = 1
      ICOUNT = 0
      SUMP   = 0.0
      P(1)   = 1.0Q0 / LONGM
      NA(1)  = LONGM
      NC(1)  = LONGM
C
C     CALCULATION OF PACKING DENSITY
      ITER   = 0
  100 ITER   = ITER + 1
      IF(L .EQ. 0)  GO TO 300
      NLONG  = 0
      NCL    = NC(L)
      DO 130    I=1,NCL
      IACMAX = 0
      DO 110    J=1,M
      IAC    = IA(J,I,L) - IC(J,1,L)
  110 IACMAX = MAX(IACMAX, ABS(IAC))

      IF(IACMAX .LT. LONG)  GO TO 130
      NLONG  = NLONG + 1
      DO 120    J=1,M
      IA(J,NLONG,L+1) = IA(J,I,L)
  120 IC(J,NLONG,L+1) = IA(J,I,L)
C
  130 CONTINUE
      IF(NLONG .EQ. 0)  GO TO 200
      L      = L + 1
      NA(L)  = NLONG
      NC(L)  = NLONG
      P(L)   = P(L-1) / NA(L)
      GO TO 100
C
  200 SUMP   = SUMP + P(L) * L
      ICOUNT = ICOUNT + 1
  210 IF(NA(L) .GE. 2)  GO TO 220
      L      = L - 1
      IF(L .EQ. 0)  GO TO 300
  220 NA(L)  = NA(L) - 1
      IF(NA(L) .EQ. 0)  GO TO 210
      NAL    = NA(L)
      DO 230    I=1,NAL
      DO 230    J=1,M
  230 IC(J,I,L) = IC(J,I+1,L)
      GO TO 100
C
  300 SUM  = SUMP / (LCUBE/LONG)**M
      WRITE(6,6300)  LCUBE, LONG, M, SUM, ICOUNT
 6300 FORMAT(1H0, '                   LENGTH OF A CUBE=', I3
     1       /1H , '            LENGTH OF PACKING CUBES=', I3
     2       /1H , '                DIMENSION OF CUBES=', I3
     3       /1H , ' EXPECTATION OF PACKING DENSITY=', 1PE15.7
     4       /1H , '             NUMBER OF COMBINATION=', I10)
C
      RETURN
      END
```

```
                 LENGTH OF A CUBE=   4
          LENGTH OF PACKING CUBES=   2
               DIMENSION OF CUBES=   3
   EXPECTATION OF PACKING DENSITY=   6.1622176Q-01
            NUMBER OF COMBINATION=     137911
```

の中から等確率でえらぶ．一般に $\vec{A}_k$ は $\vec{A}_1, \vec{A}_2, \cdots, \vec{A}_{k-1}$ とまじわらない n^m 立方体の中から等確率でえらぶものとする．このようにして n^m 立方体をつめるすきまのなくなるまでつめていく．これが $(2n)^m$ 立方体への n^m 立方体のランダムパッキングである．その結果つめられた n^m 立方体の個数の期待値をもとめる方法を考える．上記ランダムパッキングの実験において $\vec{A}_1, \vec{A}_2, \cdots, \vec{A}_N$ なる位置に n^m 立方体がランダムにつめられていったとする．$C_1, C_2, C_3, \cdots, C_N$ なる集合を考える．C_1 はなにもつまっていない状態において $\vec{A}_1$ のとりうるすべての位置からなる集合とし，C_2 は $\vec{A}_1$ が存在するという条件のもとで，つぎの n^m 立方体 $\vec{A}_2$ をつめうるすべての位置からなる集合とする．C_k は $\vec{A}_1, \vec{A}_2, \cdots, \vec{A}_{k-1}$ がつめられたという条件のもとで，つぎの n^m 立方体をつめうるすべての可能な位置の集合とする．したがって，この場合 C_{N+1} は空集合である．C_k の要素の数を $|C_k|$ と書くことにする．上記の列，$\vec{A}_1, \vec{A}_2, \cdots, \vec{A}_N$ が得られる確率 $\left(\prod_{k=1}^{N-1}|C_k|\right)^{-1}$ 及び，つめられた個数 N との積 $N\left(\prod_{k=1}^{N-1}|C_k|\right)^{-1}$ を計算する．つぎに C_N から $\vec{A}_N$ をとりのぞき，これをあらたに C_N とよぶことにする．C_N が空集合でない場合，C_N に上述のやりかたでさらにつめていく．空集合であれば $\vec{A}_{N-k}$ をとりのぞいたものが空集合にならない C_{N-k} が得られる k まで前にもどるものとする．すなわち，$\vec{A}_{N-k}$ をとりのぞいたあらたな C_{N-k} のなかへのランダムパッキングの列を考え，それをあらたに $\vec{A}_{N-k}$, $\vec{A}_{N-k+1}, \cdots, \vec{A}_{N'}$ とすることにする．このようにして得られたあらたな $\vec{A}_1, \vec{A}_2, \cdots, \vec{A}_{N'}$ 及び $C_1, C_2, \cdots, C_{N'}$ について同様に確率 $\left(\prod_{i=1}^{N'-1}|C_i|\right)^{-1}$ 及び個数との積 $N'\left(\sum_{i=1}^{N'-1}|C_i|\right)^{-1}$ を計算する．ここで $\vec{A}_1, \vec{A}_2, \cdots, \vec{A}_{N-k-1}$ 及び $C_1, C_2, \cdots, C_{N-k-1}$ は最初に考えた長さ N の列と共通である．この $\vec{A}_1, \vec{A}_2, \cdots, \vec{A}_{N'}$ 及び $C_1, C_2, \cdots, C_{N'}$ について上記と同様なことを行い，$\vec{A}_1$, $\vec{A}_2, \cdots, \vec{A}_{N''}$ 及び $C_1, C_2, \cdots, C_{N''}$ が得られる．このような操作を C_1 が空集合になるまで行う．得られた個数と確率の積の総和を 2^m でわったものが充塡率の期待値である．$n=2$ の場合について1次元から3次元まで計算した結果（表1），及び計算機プログラムを示す．

表 2　計算機実験によって求めた充塡率 $(n=2)$

次元　m	1	2	3	4	5	6	7	8	9	10	11
充塡率の標本平均 E_m	0.8348	0.7112	0.6157	0.5481	0.4927	0.4508	0.4212	0.3958	0.3762	0.3631	0.3488
充塡率の標本標準偏差	0.2352	0.2196	0.1790	0.1385	0.1044	0.0773	0.0548	0.0404	0.0277	0.0178	0.0117
実験回数	10000	10000	10000	10000	10000	10000	10000	3000	500	100	15
E_{m+1}/E_m	0.8519	0.8657	0.8902	0.8989	0.9150	0.9343	0.9397	0.9505	0.9652	0.9606	

3次元の場合，8個つめることができる．したがって上記の方法によれば少くとも $8!=40320$ 個の項を求め和をとる必要がある．$n=2$ の場合，表1にあるように137911個の項を求め和をとる必要があり，統計数理研究所の電子計算機 M200 により，26.27秒かかっている．4次元の場合16個つめることができる．したがって少くとも $16!$ 個の項を求め和をとる必要があり，計算の実行は現実的には不可能である．表1は精度34桁による計算結果である．

表2においては11次元まで計算機実験により求めた値を示す．次元があがるにつれて E_{m+1}/E_m は1に近づいて行くように見える．そうであることが証明できればおもしろいと思われる

が，これは今後明らかにすべき問題である．

謝　辞

有益な指摘をいただいた査読者の方にあつく感謝する．

参 考 文 献

[1] Akeda, Y. and Hori, M. (1976). On random sequential packing in two and three dimensions, *Biometrika*, **63**, 361-366.

[2] Bernal, J.D. and Mason, J. (1960). Co-ordination of randomly packed spheres, *Nature*, **188**, 910-911.

[3] Blaisdell, B.E. and Solomon, H. (1970). On random sequential packing in the plane and a conjecture of Palásti, *J. Appl. Prob.*, **7**, 667-698.

[4] Blaisdell, B.E. and Solomon, H. (1982). Random sequential packing in Euclidean spaces of dimensions three and four and a conjecture of Palásti, *J. Appl. Prob.*, **19**, 382-390.

[5] Dvoretzky, A. and Robbins, H. (1964). On the 'parking problem', *Publ. Math. Inst. Hung. Acad. Sci.*, **9**, 209-226.

[6] Higuti, I. (1960). A statistical study of random packing of unequal spheres, *Ann. Inst. Statist. Math.*, **12**, 257-271.

[7] Itoh, Y. (1980). On the minimum of gaps generated by one-dimensional random packing, *J. Appl. Prob.*, **17**, 134-144.

[8] Rényi, A. (1958). On a one-dimensional problem concerning space-filling, *Publ. Math. Inst. Hung. Acad. Sci.*, **3**, 109-127.

[9] Tanemura, M. (1979). On random complete packing by discs, *Ann. Inst. Statist. Math.*, **31**, 351-365.

On Packing Density by a Discrete Random Sequential Packing of Cubes in a Space of m Dimension

Yoshiaki Itoh and Sumie Ueda
(The Institute of Statistical Mathematics)

Except one dimensional case, random packing is generally very difficult to be treated theoretically. Very few is known about the packing density in case of higher dimensions. However, the difficulty could be reduced if discrete sequential packings are considered. In this article we investigate random homothetic packing of m-dimensional cubes into a larger m-dimensional cube. The expectation E_m of the packing density is estimated both by theoretical enumeration and by numerical experiment for a simple case. Imagine the m-dimensional cubic lattice with unit lattice vector in an m-dimensional cube of side length $2n$. The vertices of the cube are also lattice points. m-dimensional cubes of side length n are put sequentially into the larger cube so as each vertex coincides with one of the lattice points. The position of the smaller cubes in chosen at random unless the cube extrudes out of the larger cube or overlaps the cubes already put. In practice, the numerical calculation of E_m is very complicated and needs a lot of computational time, so we can do for only first few dimensions; up to the case of three dimension. The results are shown in Table 1. The estimation of E_m by experiment was executed up to 11 dimension and the results are shown in Table 2. It seems that E_{m+1}/E_m increases with the dimension m and tends to 1 as m becomes sufficiently large.